Blatt 11

Aufgabe 42

```
> restart;
> with(LinearAlgebra):
> f := (x, y, z) -> x^2 - y^2 + z^2 - (x^2 + 2*y^2 +4*z^2)^2;
```

$$f := (x, y, z) \mapsto x^2 - y^2 + z^2 - \left(x^2 + 2\,y^2 + 4\,z^2\right)^2 \quad \textbf{(1.1)}$$

```
> Df := [ diff(f(x, y, z), x), diff(f(x, y, z), y), diff(f(x, y,
  z), z) ];
```

$$Df := \left[2\,x - 4\left(x^2 + 2\,y^2 + 4\,z^2\right)x, \ \text{-}2\,y - 8\left(x^2 + 2\,y^2 + 4\,z^2\right)y, \ 2\,z - 16\left(x^2 + 2\,y^2 + 4\,z^2\right)z\right] \quad \textbf{(1.2)}$$

```
> D2f := < < diff(f(x, y, z), x$2), diff(f(x, y, z), [ x, y ]),
  diff(f(x, y, z), [ x, z ]) > |
  < diff(f(x, y, z), [ y, x ]), diff(f(x, y, z), y$2), diff(f(x,
  y, z), [ y, z ]) > |
  < diff(f(x, y, z), [ z, x ]), diff(f(x, y, z), [ z, y ]), diff
  (f(x, y, z), z$2) > >;
```

$$D2f := \left[\left[\text{-}12\,x^2 - 8\,y^2 - 16\,z^2 + 2, \ \text{-}16\,y\,x, \ \text{-}32\,z\,x\right], \left[\text{-}16\,y\,x, \ \text{-}8\,x^2 - 48\,y^2 - 32\,z^2 - 2, \ \text{-}64\,z\,y\right], \left[\text{-}32\,z\,x, \ \text{-}64\,z\,y, \ \text{-}16\,x^2 - 32\,y^2 - 192\,z^2 + 2\right]\right] \quad \textbf{(1.3)}$$

```
> kritischePunkte := solve([ seq(Df[i] = 0, i = 1..3) ], [ x, y,
  z ]):
> kritischePunkte := seq(allvalues(kritischePunkte[kk]), kk = 1..
  nops(kritischePunkte));
```

$$kritischePunkte := [x = 0, y = 0, z = 0], \left[x = \frac{\sqrt{2}}{2}, y = 0, z = 0\right], \left[x = \text{-}\frac{\sqrt{2}}{2}, y = 0, z = 0\right], \left[x = 0, y = \frac{I}{4}\sqrt{2}, z = 0\right], \left[x = 0, y = \text{-}\frac{I}{4}\sqrt{2}, z = 0\right], \left[x = 0, y = 0, z = \frac{\sqrt{2}}{8}\right], \left[x = 0, y = 0, z = \text{-}\frac{\sqrt{2}}{8}\right] \quad \textbf{(1.4)}$$

```
> # Reelle Lösungen
> kritischePunkte := [ seq(kritischePunkte[kk], kk in [ 1, 2, 3,
  6, 7 ]) ];
```

$$kritischePunkte := \left[[x = 0, y = 0, z = 0], \left[x = \frac{\sqrt{2}}{2}, y = 0, z = 0\right], \left[x = \text{-}\frac{\sqrt{2}}{2}, y = 0, z = 0\right], \left[x = 0, y = 0, z = \frac{\sqrt{2}}{8}\right], \left[x = 0, y = 0, z = \text{-}\frac{\sqrt{2}}{8}\right]\right] \quad \textbf{(1.5)}$$

```
> # Prüfe Definitheit der Hesse-Matrix
> seq(print(kritischePunkte[kk], 'EW' = Eigenvalues(subs
```

```
(kritischePunkte[kk], D2f))), kk = 1..nops(kritischePunkte));
```

$$[x = 0,\ y = 0,\ z = 0],\ EW = \begin{bmatrix} -2 \\ 2 \\ 2 \end{bmatrix}$$

$$\left[x = \frac{\sqrt{2}}{2},\ y = 0,\ z = 0\right],\ EW = \begin{bmatrix} -4 \\ -6 \\ -6 \end{bmatrix}$$

$$\left[x = -\frac{\sqrt{2}}{2},\ y = 0,\ z = 0\right],\ EW = \begin{bmatrix} -4 \\ -6 \\ -6 \end{bmatrix}$$

$$\left[x = 0,\ y = 0,\ z = \frac{\sqrt{2}}{8}\right],\ EW = \begin{bmatrix} -4 \\ -3 \\ \frac{3}{2} \end{bmatrix}$$

$$\left[x = 0,\ y = 0,\ z = -\frac{\sqrt{2}}{8}\right],\ EW = \begin{bmatrix} -4 \\ -3 \\ \frac{3}{2} \end{bmatrix} \quad \textbf{(1.6)}$$

```
> # Das heißt: 0 und (0,0,+-sqrt(2)/8) sind Sattelpunkte,
  # die anderen beiden lokale Maxima
> plot3d(f(x, 0, y), x = -1..1, y = -1/2..1/2, style =
  patchcontour, contours=35, view = -0.3 .. 0.3, numpoints =
  3000);
```




```
> plot3d(f(x, y, 0), x = -1..1, y = -1/2..1/2, style =
  patchcontour, contours=35, view = -0.3 .. 0.3, numpoints =
  3000);
```



Aufgabe 43

```
> restart;
> with(LinearAlgebra):
> with(VectorCalculus):
> with(plots):
> q := < x, y, z >;
```

$$q := (x)e_x + (y)e_y + (z)e_z \quad (2.1)$$

```
> q1 := < 1, 0, 0 >;
> q2 := < 0, 1, 0 >;
```

$$q1 := e_x$$

$$q2 := e_y \quad (2.2)$$

```
> f := 3 / VectorNorm(q1 - q, 2) + 2 / VectorNorm(q2 - q, 2);
> Gradf := Gradient(f, [ x, y, z ]);
```

$$f := \frac{3}{\sqrt{|x-1|^2 + |y|^2 + |z|^2}} + \frac{2}{\sqrt{|x|^2 + |y-1|^2 + |z|^2}}$$

$$Gradf := \left(-\frac{3\,|x-1|\,\mathrm{abs}(1,x-1)}{(|x-1|^2+|y|^2+|z|^2)^{3/2}} - \frac{2\,|x|\,\mathrm{abs}(1,x)}{(|x|^2+|y-1|^2+|z|^2)^{3/2}}\right)\bar{e}_x + \left(-\frac{3\,|y|\,\mathrm{abs}(1,y)}{(|x-1|^2+|y|^2+|z|^2)^{3/2}} - \frac{2\,|y-1|\,\mathrm{abs}(1,y-1)}{(|x|^2+|y-1|^2+|z|^2)^{3/2}}\right)\bar{e}_y + \left(-\frac{3\,|z|\,\mathrm{abs}(1,z)}{(|x-1|^2+|y|^2+|z|^2)^{3/2}} - \frac{2\,|z|\,\mathrm{abs}(1,z)}{(|x|^2+|y-1|^2+|z|^2)^{3/2}}\right)\bar{e}_z \quad (2.3)$$

```
> fieldplot3d(Gradf, x = -0.5..1.5, y = -0.5..1.5, z = -1..1,
  thickness = 2, axes = boxed);
```



Aufgabe 44

```
> restart;
> f := x -> -(x^3 + 3*x^2 + 4*x + 3) * exp(-x);
```

$$f := x \mapsto -(x^3+3\,x^2+4\,x+3)\,\mathrm{e}^{-x} \quad (3.1)$$

```
> df := D(f);
```

$$df := x \mapsto -(3\,x^2+6\,x+4)\,\mathrm{e}^{-x} + (x^3+3\,x^2+4\,x+3)\,\mathrm{e}^{-x} \quad (3.2)$$

```
> d2f := D(df);
```

$$d2f := x \mapsto -(6\,x+6)\,e^{-x} + 2\,(3\,x^2+6\,x+4)\,e^{-x} - (x^3+3\,x^2+4\,x+3)\,e^{-x} \tag{3.3}$$

(a)

```
> plot([ f(x), df(x), d2f(x) ],
      x = -2..6, y = -5..5,
      color = [ blue, red, magenta ], legend = [ 'f', "f'", "f''"
  ],
     numpoints = 1000);
```



(b)

```
> minimize(f(x), x = -1..2);
```

$$-\left(\left(\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^3 + 3\left(\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^2 + 2\sqrt{5} + 5\right) e^{-\frac{\sqrt{5}}{2}-\frac{1}{2}} \tag{3.4}$$

```
> maximize(f(x), x = -1..2);
```

$$-\left(\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^3 + 3\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^2 + 5 - 2\sqrt{5}\right) e^{-\frac{1}{2}+\frac{\sqrt{5}}{2}} \tag{3.5}$$

(c)

```
> kritischeStellen := solve({ df(x) = 0 }, { x });
```

$$kritischeStellen := \{x=-1\}, \left\{x=\frac{1}{2}-\frac{\sqrt{5}}{2}\right\}, \left\{x=\frac{\sqrt{5}}{2}+\frac{1}{2}\right\} \quad (3.6)$$

```
> #evalf(kritischeStellen) ;
> rhs(kritischeStellen[1][1]);
```

$$-1 \quad (3.7)$$

```
> f(rhs(kritischeStellen[2][1]));
```

$$-\left(\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^3+3\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^2+5-2\sqrt{5}\right)e^{-\frac{1}{2}+\frac{\sqrt{5}}{2}} \quad (3.8)$$

```
> mm[false] := "Maximum"; mm[true] := "Minimum";
```

$$mm_{false} := \text{"Maximum"}$$

$$mm_{true} := \text{"Minimum"} \quad (3.9)$$

```
> seq(print([ f(rhs(x[1])), d2f(rhs(x[1])), mm[is(d2f(rhs(x[1])),
  positive)] ]), x in [ kritischeStellen ]);
```

$$[-e, e, \text{"Minimum"}]$$

$$\left[-\left(\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^3+3\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^2+5-2\sqrt{5}\right)e^{-\frac{1}{2}+\frac{\sqrt{5}}{2}}, -\left(9-3\sqrt{5}\right)e^{-\frac{1}{2}+\frac{\sqrt{5}}{2}}+2\left(3\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^2+7-3\sqrt{5}\right)e^{-\frac{1}{2}+\frac{\sqrt{5}}{2}}-\left(\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^3+3\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^2+5-2\sqrt{5}\right)e^{-\frac{1}{2}+\frac{\sqrt{5}}{2}}, \text{"Maximum"}\right]$$

$$\left[-\left(\left(\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^3+3\left(\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^2+2\sqrt{5}+5\right)e^{-\frac{\sqrt{5}}{2}-\frac{1}{2}}, -\left(3\sqrt{5}+9\right)e^{-\frac{\sqrt{5}}{2}-\frac{1}{2}}+2\left(3\left(\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^2+3\sqrt{5}+7\right)e^{-\frac{\sqrt{5}}{2}-\frac{1}{2}}-\left(\left(\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^3+3\left(\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^2+2\sqrt{5}+5\right)e^{-\frac{\sqrt{5}}{2}-\frac{1}{2}}, \text{"Minimum"}\right] \quad (3.10)$$

(d)

```
> tangente := x -> f(0) + (x - 0) * df(0);
```

$$tangente := x \mapsto f(0)+x\,df(0) \quad (3.11)$$

```
> plot([ f(x), tangente(x) ], x = -2..2);
```




```
> schnittpunkte := solve({ tangente(x) = f(x) }, x);
> schnittpunkte := allvalues(schnittpunkte);
> evalf(schnittpunkte); # Hmmmm
```

Warning, solutions may have been lost

$$schnittpunkte := \{x = RootOf(e^{-\_Z}\_Z^3 + 3\,e^{-\_Z}\_Z^2 + 4\,e^{-\_Z}\_Z + 3\,e^{-\_Z} - \_Z - 3)\}$$

$$schnittpunkte := \{x = RootOf(e^{-\_Z}\_Z^3 + 3\,e^{-\_Z}\_Z^2 + 4\,e^{-\_Z}\_Z + 3\,e^{-\_Z} - \_Z - 3, -1.532145884)\}, \{x = RootOf(e^{-\_Z}\_Z^3 + 3\,e^{-\_Z}\_Z^2 + 4\,e^{-\_Z}\_Z + 3\,e^{-\_Z} - \_Z - 3, 0.)\}$$

$$\{x = -1.532145884\}, \{x = 0.\} \quad \textbf{(3.12)}$$

```
> s1 := fsolve({ tangente(x) = f(x) }, x);
> s1 := rhs(s1[1]);
```

$$s1 := \{x = 0.\}$$

$$s1 := 0. \quad \textbf{(3.13)}$$

```
> s2 := fsolve({ tangente(x) = f(x) }, x, avoid = { x = s1 });
> s2 := rhs(s2[1]);
```

$$s2 := \{x = -1.532145884\}$$
$$s2 := -1.532145884 \quad \textbf{(3.14)}$$

```
> s3 := fsolve({ tangente(x) = f(x) }, x, avoid = { x = s1, x = s2 });
> s3 := rhs(s3[1]);
```

$$s3 := \{x = 1.136613580\}$$
$$s3 := 1.136613580 \quad \textbf{(3.15)}$$

Aufgabe 45

```
> restart;
> with(VectorCalculus):
> SetCoordinates('cartesian'[x[1], x[2], x[3]]);
```

$$cartesian_{x_1, x_2, x_3} \quad \textbf{(4.1)}$$

```
> BasisFormat(false);
```

$$true \quad \textbf{(4.2)}$$

```
> f := x -> exp(x[1] * x[2]) * arctan(x[2] * x[3]);
```

$$f := x \mapsto e^{x_1 x_2} \arctan(x_2 x_3) \quad \textbf{(4.3)}$$

(a)

```
> df := x -> < diff(f(x), x[1]), diff(f(x), x[2]), diff(f(x), x[3]) >;
> gr := Gradient(f(x), [ x[1], x[2], x[3] ]);
> # Test
> VectorField(df(x)) - gr;
```

$$df := x \mapsto \left\langle \frac{\partial}{\partial x_1} f(x), \frac{\partial}{\partial x_2} f(x), \frac{\partial}{\partial x_3} f(x) \right\rangle$$

$$gr := \begin{bmatrix} x_2 e^{x_1 x_2} \arctan(x_2 x_3) \\ x_1 e^{x_1 x_2} \arctan(x_2 x_3) + \frac{e^{x_1 x_2} x_3}{x_2^2 x_3^2 + 1} \\ \frac{e^{x_1 x_2} x_2}{x_2^2 x_3^2 + 1} \end{bmatrix}$$

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \quad \textbf{(4.4)}$$

```
> d2f := x -> < < diff(f(x), x[1]$2),    diff(f(x), x[2], x[1]), diff(f(x), x[3], x[1]) > |
```

```
              < diff(f(x), x[1], x[2]), diff(f(x), x[2]$2),
  diff(f(x), x[3], x[2]) > |
              < diff(f(x), x[1], x[3]), diff(f(x), x[2], x[3]),
  diff(f(x), x[3]$2)    > >;
> H := Hessian(f(x), [ x[1], x[2], x[3] ]);
> # Test
> d2f(x) - H;
```

$$d2f := x \mapsto \left\langle \left\langle \frac{\partial^2}{\partial x_1^2} f(x), \frac{\partial^2}{\partial x_2 \partial x_1} f(x), \frac{\partial^2}{\partial x_3 \partial x_1} f(x) \right\rangle \middle| \left\langle \frac{\partial^2}{\partial x_1 \partial x_2} f(x), \frac{\partial^2}{\partial x_2^2} f(x), \frac{\partial^2}{\partial x_3 \partial x_2} f(x) \right\rangle \middle| \left\langle \frac{\partial^2}{\partial x_1 \partial x_3} f(x), \frac{\partial^2}{\partial x_2 \partial x_3} f(x), \frac{\partial^2}{\partial x_3^2} f(x) \right\rangle \right\rangle$$

$$H := \left[\left[x_2^2 e^{x_1 x_2} \arctan(x_2 x_3), e^{x_1 x_2} \arctan(x_2 x_3) + x_2 x_1 e^{x_1 x_2} \arctan(x_2 x_3) + \frac{x_2 e^{x_1 x_2} x_3}{x_2^2 x_3^2 + 1}, \frac{x_2^2 e^{x_1 x_2}}{x_2^2 x_3^2 + 1} \right], \right.$$
$$\left[e^{x_1 x_2} \arctan(x_2 x_3) + x_2 x_1 e^{x_1 x_2} \arctan(x_2 x_3) + \frac{x_2 e^{x_1 x_2} x_3}{x_2^2 x_3^2 + 1}, x_1^2 e^{x_1 x_2} \arctan(x_2 x_3) + \frac{2 x_1 e^{x_1 x_2} x_3}{x_2^2 x_3^2 + 1} - \frac{2 e^{x_1 x_2} x_3^3 x_2}{(x_2^2 x_3^2 + 1)^2}, \frac{x_1 e^{x_1 x_2} x_2}{x_2^2 x_3^2 + 1} + \frac{e^{x_1 x_2}}{x_2^2 x_3^2 + 1} - \frac{2 e^{x_1 x_2} x_3^2 x_2^2}{(x_2^2 x_3^2 + 1)^2} \right],$$
$$\left. \left[\frac{x_2^2 e^{x_1 x_2}}{x_2^2 x_3^2 + 1}, \frac{x_1 e^{x_1 x_2} x_2}{x_2^2 x_3^2 + 1} + \frac{e^{x_1 x_2}}{x_2^2 x_3^2 + 1} - \frac{2 e^{x_1 x_2} x_3^2 x_2^2}{(x_2^2 x_3^2 + 1)^2}, -\frac{2 e^{x_1 x_2} x_2^3 x_3}{(x_2^2 x_3^2 + 1)^2} \right] \right]$$

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad \textbf{(4.5)}$$

(b)

```
> r := (s, x) -> f(<x[1], x[2], x[3]> + <1, 1, 1> * s);
```

$$r := (s, x) \mapsto f(\langle x_1, x_2, x_3 \rangle + (\langle 1, 1, 1 \rangle s)) \quad \textbf{(4.6)}$$

```
> ra := unapply(eval(diff(r(s, x), s), s = 0), x);
```

$$ra := x \mapsto (x_2 + x_1) e^{x_1 x_2} \arctan(x_2 x_3) + \frac{e^{x_1 x_2} (x_3 + x_2)}{x_2^2 x_3^2 + 1} \quad \textbf{(4.7)}$$

```
> # Alternative Definition
> ra_Grad := gr . <1,1,1>;
```

$$ra\_Grad := x_2\, e^{x_1 x_2} \arctan(x_2 x_3) + x_1\, e^{x_1 x_2} \arctan(x_2 x_3) + \frac{e^{x_1 x_2} x_3}{x_2^2 x_3^2 + 1} + \frac{e^{x_1 x_2} x_2}{x_2^2 x_3^2 + 1} \quad (4.8)$$

```
> # Test
> simplify(ra_Grad - ra(x));
```

$$0 \quad (4.9)$$

(c)

```
> points := <0, 1, 0>, <1, 1, 1>;
```

$$points := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \quad (4.10)$$

```
> seq(print(ra(p)), p in points);
```

$$1$$

$$\frac{e\,\pi}{2} + e \quad (4.11)$$